The Gauss elimination game and applications to systems of DEs

Prof. Joyner[1]

This is actually a lecture on solving systems of equations using the method of *row reduction*, but it's more fun to formulate it in terms of a game.

To be specific, let's focus on a 2×2 system (by "2×2" I mean 2 equations in the 2 unknowns x, y):

$$\begin{cases} ax + by & = & r_1 \\ cx + dy & = & r_2 \end{cases} \tag{1}$$

Here a, b, c, d, r_1, r_2 are given constants. Putting these two equations down together means to solve them simultaneously, not individually. In geometric terms, you may think of each equation above as a line the the plane. To solve them simultaneously, you are to find the point of intersection (if it exists) of these two lines. Since a, b, c, d, r_1, r_2 have not been specified, it is conceivable that there are

- no solutions (the lines are parallel but distinct),
- infinitely many solutions (the lines are the same),
- exactly one solution (the lines are distinct and not parallel).

"Usually" there is exactly one solution. Of course, you can solve this by simply manipulating equations since it is such a low-dimensional system but the object of this lecture is to show you a method of solution which is "scalable" to "industrial-sized" problems (say 1000×1000 or larger).

Strategy:

Step 1: Write down the *augmented matrix* of (1):

$$A = \begin{pmatrix} a & b & r_1 \\ c & d & r_2 \end{pmatrix}$$

This is simply a matter of stripping off the unknowns and recording the coefficients in an array. Of course, the system must be written in "standard form" (all the terms with "x" get aligned together, ...) to do this correctly.
Step 2: Play the Gauss elimination game (described below) to computing the row reduced echelon form of A, call it B say.
Step 3: Read off the solution from the right-most column of B.

The Gauss Elimination Game

Legal moves: These actually apply to any $m \times n$ matrix A with $m < n$.

1. $R_i \leftrightarrow R_j$: You can swap row i with row j.
2. $cR_i \to R_i$ ($c \neq 0$): You can replace row i with row i multiplied by any non-zero constant c. (Don't confuse this c with the c in (1)).
3. $cR_i + R_j \to R_i$ ($c \neq 0$): You can replace row i with row i multiplied by any non-zero constant c plus row j, $j \neq i$.

Note that move 1 simply corresponds to reordering the system of equations (1)). Likewise, move 2 simply corresponds to scaling equation i in (1)). In general, these "legal moves" correspond to algebraic operations you would perform on (1)) to solve it. However, there are fewer symbols to push around when the augmented matrix is used.

Goal: You *win* the game when you can achieve the following situation. Your goal is to find a sequence of legal moves leading to a matrix B satisfying the following criteria:

1. all rows of B have leaading non-zero term equal to 1 (the position where this leading term in B occurs is called a *pivot position*),
2. B contains as many 0's as possible
3. all entries above and below a pivot position must be 0,
4. the pivot position of the i^{th} row is to the left and above the pivot position of the $(i+1)^{st}$ row (therefore, all entries below the diagonal of B are 0),
5. each of the all-zeros rows (if any) must be at the bottom.

This matrix B is unique (this is a theorem which you can find in any text on elementary matrix theory or linear algebra[2]) and is called the *row reduced echelon form* of A, sometimes written $rref(A)$.

Two comments: (1) If you are your friend both start out playing this game, it is likely your choice of legal moves will differ. That is to be expected. However, you must get the same result in the end. (2) Often if someone is to get "stuck" it is becuase they forget that one of the goals is to "kill as many terms as possible (i.e., you need B to have as many 0's as possible). If you forget this you might create non-zero terms in the matrix while killing others. You should try to think of each move as being made in order to to kill a term. The exception is at the very end where you can't kill any more terms but you want to do row swaps to put it in diagonal form.

Now it's time for an example.

Example: Solve

$$\left\{ \begin{array}{rcl} x + 2y &=& 3 \\ 4x + 5y &=& 6 \end{array} \right. \tag{2}$$



Figure 1: lines $x + 2y = 3$, $4x + 5y = 6$ in the plane.

The augmented matrix is

$$A = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$$

[2]For example, [B] or [H].

One sequence of legal moves is the following:

$-4R_1 + R_2 \to R_2$, which leads to $\begin{pmatrix} 1 & 2 & 3 \\ 0 & -3 & -6 \end{pmatrix}$

$-(1/3)R_2 \to R_2$, which leads to $\begin{pmatrix} 1 & 2 & 3 \\ 0 & 1 & 2 \end{pmatrix}$

$-2R_2 + R_1 \to R_1$, which leads to $\begin{pmatrix} 1 & 0 & -1 \\ 0 & 1 & 2 \end{pmatrix}$

Now we are done (we won!) since this matrix satisfies all the goals for a eow reduced echelon form.

The latter matrix corresponds to the system of equations

$$\left\{ \begin{array}{rcr} x + 0y & = & -1 \\ 0x + y & = & 2 \end{array} \right. \tag{3}$$

Since the "legal moves" were simply matrix analogs of algebraic manipulations you'd appy to the system (2), the solution to (2) is the same as the solution to (3), whihc is obviously $x = -1, y = 2$. You can visually check this from the graph given above.

To find the row reduced echelon form of

$$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$$

using SAGE, just type the following:

SAGE

```
sage: MS = MatrixSpace(QQ,2,3)
sage: A = MS([[1,2,3],[4,5,6]])
sage: A
[1 2 3]
[4 5 6]
sage: A.echelon_form()
[ 1  0 -1]
[ 0  1  2]
```

Solving systems using inverses

There is another method of solving "square" systems of linear equations which we discuss next.

One can rewrite the system (1) as a single matrix equation

$$\begin{pmatrix} a & b \\ c & d \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} r_1 \\ r_2 \end{pmatrix},$$

or more compactly as

$$A\vec{X} = \vec{r}, \tag{4}$$

where $\vec{X} = \begin{pmatrix} x \\ y \end{pmatrix}$ and $\vec{r} = \begin{pmatrix} r_1 \\ r_2 \end{pmatrix}$. How do you solve (4)? The obvious this to do ("divide by A") is the right idea:

$$\begin{pmatrix} x \\ y \end{pmatrix} = \vec{X} = A^{-1}\vec{r}.$$

Here A^{-1} is a matrix with the property that $A^{-1}A = I$, the identity matrix (which satisfies $I\vec{X} = \vec{X}$).

If A^{-1} exists (and it usually does), how do we compute it? There are a few ways. One, if using a formula. In the 2×2 case, the inverse is given by

$$\begin{pmatrix} a & b \\ c & d \end{pmatrix}^{-1} = \frac{1}{ad - bc} \begin{pmatrix} d & -b \\ -c & a \end{pmatrix}.$$

There is a similar formula for larger sized matrices but it is so unwieldy that is is usually not used to compute the inverse. In the 2×2 case, it is easy to use and we see for example,

$$\begin{pmatrix} 1 & 2 \\ 4 & 5 \end{pmatrix}^{-1} = \frac{1}{-3} \begin{pmatrix} 5 & -2 \\ -4 & 1 \end{pmatrix} = \begin{pmatrix} -5/3 & 2/3 \\ 4/3 & -1/3 \end{pmatrix}.$$

To find the inverse of

$$\begin{pmatrix} 1 & 2 \\ 4 & 5 \end{pmatrix}$$

using SAGE, just type the following:

SAGE

```
sage: MS = MatrixSpace(QQ,2,2)
sage: A = MS([[1,2],[4,5]])
sage: A
[1 2]
```

```
[4 5]
sage: A^(-1)
[-5/3  2/3]
[ 4/3 -1/3]
```

A better way to compute A^{-1} is the following. Compute the row reduced echelon form of the matrix (A, I), where I is the identity matrix of the same size as A. This new matrix will be (if the inverse exists) (I, A^{-1}). You can read off the inverse matrix from this.

Here is an example.

Example Solve

$$\begin{cases} x + 2y &= 3 \\ 4x + 5y &= 6 \end{cases}$$

using matrix inverses.

This is

$$\begin{pmatrix} 1 & 2 \\ 4 & 5 \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 3 \\ 6 \end{pmatrix},$$

so

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 1 & 2 \\ 4 & 5 \end{pmatrix}^{-1} \begin{pmatrix} 3 \\ 6 \end{pmatrix}.$$

To compute the inverse matrix, apply the Gauss elimination game to

$$\begin{pmatrix} 1 & 2 & 1 & 0 \\ 4 & 5 & 0 & 1 \end{pmatrix}$$

Using the same sequence of legal moves as in the previous example, we get

$-4R_1 + R_2 \to R_2$, which leads to $\begin{pmatrix} 1 & 2 & 1 & 0 \\ 0 & -3 & -4 & 1 \end{pmatrix}$

$-(1/3)R_2 \to R_2$, which leads to $\begin{pmatrix} 1 & 2 & 1 & 0 \\ 0 & 1 & 4/3 & -1/3 \end{pmatrix}$

$-2R_2 + R_1 \to R_1$, which leads to $\begin{pmatrix} 1 & 0 & -5/3 & 2/3 \\ 0 & 1 & 4/3 & -1/3 \end{pmatrix}$.

Therefore the inverse is

$$A^{-1} = \begin{pmatrix} -5/3 & 2/3 \\ 4/3 & -1/3 \end{pmatrix}.$$

Now, to solve the system, compute

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 1 & 2 \\ 4 & 5 \end{pmatrix}^{-1} \begin{pmatrix} 3 \\ 6 \end{pmatrix} = \begin{pmatrix} -5/3 & 2/3 \\ 4/3 & -1/3 \end{pmatrix} \begin{pmatrix} 3 \\ 6 \end{pmatrix} = \begin{pmatrix} -1 \\ 2 \end{pmatrix}.$$

To make SAGE do the above computation, just type the following:

SAGE

```
sage: MS = MatrixSpace(QQ,2,2)
sage: A = MS([[1,2],[4,5]])
sage: V = VectorSpace(QQ,2)
sage: v = V([3,6])
sage: A^(-1)*v
 (-1, 2)
```

Finding matrix kernels using row reduction

The kernel of a matrix A is the set of all vectors v such that "A kills v": $Av = 0$. (If A is a square matrix, the only time you will get a non-zero solution to $Av = 0$ is when A is singular, ie, $\det(A) = 0$.) Of course, this is just a special case of solving a linear system of equations.

Here is an example.

Example Solve $Av = \lambda v$ where

$$A = \begin{pmatrix} 0 & 4 & 0 \\ -1 & -4 & 0 \\ 0 & 0 & -2 \end{pmatrix}$$

$v = {}^t(x, y, z)$ and $\lambda = -2$. This is the kernel of $A - \lambda I$. The corresponding system is

$$\begin{cases} 2x + 4y &= 0 \\ -x - 2y &= 0 \end{cases}$$

(and there is no constraint on z). Using elementary row operations, we obtain in a few steps

$$\text{rref}\begin{pmatrix} 2 & 4 & 0 & 0 \\ -1 & -2 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} = \begin{pmatrix} 1 & 2 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}.$$

This imposes only the constraint $x + 2y = 0$, so

$$\ker(A) = \{ \,{}^t(x, y, z) \mid x + 2y = 0\} = \langle \,{}^t(0, 0, 1), \ {}^t(-2, 1, 0) \,\rangle,$$

where $\langle ... \rangle$ denotes the span.

Application: Solving systems of DEs

Suppose we have a system of DEs in "standard form"

$$\begin{cases} x' &= ax + by + f(t), \quad x(0) = x_0, \\ y' &= cx + dy + g(t), \quad y(0) = y_0, \end{cases} \tag{5}$$

where a, b, c, d, x_0, y_0 are given constants and $f(t), g(t)$ are given "nice" functions. (Here "nice" will be left vague but basically we don't want these functions to annoy us with any bad behaviour while trying to solve the DEs by the method of Laplace transforms.)

One way to solve this system if to take Laplace transforms of both sides. If we let

$$X(s) = \mathcal{L}[x(t)](s), Y(s) = \mathcal{L}[y(t)](s), F(s) = \mathcal{L}[f(t)](s), G(s) = \mathcal{L}[g(t)](s),$$

then (5) becomes

$$\begin{cases} sX(s) - x_0 &= aX(s) + bY(s) + F(s), \\ sY(s) - y_0 &= cX(s) + dY(s) + G(s). \end{cases} \tag{6}$$

This is now a 2×2 system of linear equations in the unknowns $X(s), Y(s)$ with augmented matrix

$$A = \begin{pmatrix} s-a & -b & F(s) + x_0 \\ -c & s-d & G(s) + y_0 \end{pmatrix}.$$

Example: Solve

$$\begin{cases} x' &= -y + 1, \quad x(0) = 0, \\ y' &= -x + t, \quad y(0) = 0, \end{cases}$$

The augmented matrix is

$$A = \begin{pmatrix} s & 1 & 1/s \\ 1 & s & 1/s^2 \end{pmatrix}.$$

The row reduced echelon form of this is

$$\begin{pmatrix} 1 & 0 & 1/s^2 \\ 0 & 1 & 0 \end{pmatrix}.$$

Therefore, $X(s) = 1/s^2$ and $Y(s) = 0$. Taking inverse Laplace transforms, we see that the solution to the system is $x(t) = t$ and $y(t) = 0$. It is easy to check that this is indeed the solution.

To make SAGE compute the row reduced echelon form, just type the following:

SAGE

```
sage: R = PolynomialRing(QQ,"s")
sage: F = FractionField(R)
sage: s = F.gen()
sage: MS = MatrixSpace(F,2,3)
sage: A = MS([[s,1,1/s],[1,s,1/s^2]])
sage: A.echelon_form()
[    1     0 1/s^2]
[    0     1     0]
```

To make SAGE compute the Laplace transform, just type the following:

SAGE

```
sage: maxima("laplace(1,t,s)")
 1/s
sage: maxima("laplace(t,t,s)")
 1/s^2
```

To make SAGE compute the inverse Laplace transform, just type the following:

SAGE

```
sage: maxima("ilt(1/s^2,s,t)")
```

```
 t
sage: maxima("ilt(1/(s^2+1),s,t)")
 sin(t)
```

Example: Solve

$$\left\{ \begin{array}{lll} x' & = & -4y, \quad x(0)=400, \\ y' & = & -x, \quad y(0)=100, \end{array} \right.$$

This models a battle between "x-men" and "y-men", where the "x-men" die off at a higher rate than the "y-men" (but there are more of them to begin with too).

The augmented matrix is

$$A = \left(\begin{array}{ccc} s & 4 & 400 \\ 1 & s & 100 \end{array} \right).$$

The row reduced echelon form of this is

$$\left(\begin{array}{ccc} 1 & 0 & \frac{400(s-1)}{s^2-4} \\ 0 & 1 & \frac{100(s-4)}{s^2-4} \end{array} \right).$$

Therefore,

$$X(s) = 400\frac{s}{s^2-4} - 200\frac{2}{s^2-4}, \quad Y(s) = 100\frac{s}{s^2-4} - 200\frac{2}{s^2-4}.$$

Taking inverse Laplace transforms, we see that the solution to the system is $x(t) = 400\cosh(2t) - 200\sinh(2t)$ and $y(t) = 100\cosh(2t) - 200\sinh(2t)$. The "$x$-men" win and, in fact,

$$x(0.275) = 346.4102..., \quad y(0.275) = -0.1201... \; .$$

Question: What is $x(t)^2 - 4y(t)^2$? (Hint: It's a constant. Can you explain this?)

To make SAGE plot this just type the following:

SAGE

```
sage: f = lambda x: 400*cosh(2*x)-200*sinh(2*x)
sage: g = lambda x: 100*cosh(2*x)-200*sinh(2*x)
sage: P = plot(f,0,1)
sage: Q = plot(g,0,1)
sage: show(P+Q)
sage: g(0.275)
 -0.12017933629675781
sage: f(0.275)
 346.41024490088557
```



Figure 2: curves $x(t) = 400\cosh(2t) - 200\sinh(2t)$, $y(t) = 100\cosh(2t) - 200\sinh(2t)$ along the t-axis.

Example: The displacement from equilibrium (respectively) for coupled springs attached to a wall on the left

coupled springs

```
|------\/\/\/\/\---|mass1|----\/\/\/\/\/----|mass2|
         spring1                spring2
```

is modeled by the system of 2nd order ODEs

$$m_1 x_1'' + (k_1 + k_2)x_1 - k_2 x_2 = 0, \quad m_2 x_2'' + k_2(x_2 - x_1) = 0,$$

where x_1 denotes the displacement from equilibrium of mass 1, denoted m_1, x_2 denotes the displacement from equilibrium of mass 2, denoted m_2, and k_1, k_2 are the respective spring constants [CS].

As another illustration of solving linear systems of equations to solving systems of linear 1st order DEs, we use SAGE to solve the above problem with $m_1 = 2$, $m_2 = 1$, $k_1 = 4$, $k_2 = 2$, $x_1(0) = 3$, $x_1'(0) = 0$, $x_2(0) = 3$, $x_2'(0) = 0$.

Soln: Take Laplace transforms of the first DE (for simplicity of notation, let $x = x_1$, $y = x_2$):

SAGE+Maxima

```
sage: _ = maxima.eval("x2(t) := diff(x(t),t, 2)")
sage: maxima("laplace(2*x2(t)+6*x(t)-2*y(t),t,s)")
2*(-?%at('diff(x(t),t,1),t=0)+s^2*?%laplace(x(t),t,s)-x(0)*s)-2*?%laplace(y(t),t,s)+6*?%laplace(x(t),t,s)
```

This says $-2x_1'(0) + 2s^2 * X_1(s) - 2sx_1(0) - 2X_2(s) + 2X_1(s) = 0$ (where the Laplace transform of a lower case function is the upper case function). Take Laplace transforms of the second DE:

SAGE+Maxima

```
sage: _ = maxima.eval("y2(t) := diff(y(t),t, 2)")
sage: maxima("laplace(y2(t)+2*y(t)-2*x(t),t,s)")
-?%at('diff(y(t),t,1),t=0)+s^2*?%laplace(y(t),t,s)+2*?%laplace(y(t),t,s)-2*?%laplace(x(t),t,s)-y(0)*s
```

This says $s^2 X_2(s) + 2X_2(s) - 2X_1(s) - 3s = 0$. Solve these two equations:

SAGE

```
sage: s,X,Y = var('s X Y')
```

```
sage: eqns = [(2*s^2+6)*X-2*Y == 6*s, -2*X +(s^2+2)*Y == 3*s]
sage: solve(eqns, X,Y)
[[X == (3*s^3 + 9*s)/(s^4 + 5*s^2 + 4),
  Y == (3*s^3 + 15*s)/(s^4 + 5*s^2 + 4)]]
```

This says $X_1(s) = (3s^3+9s)/(s^4+5s^2+4)$, $X_2(s) = (3s^3+15s)/(s^4+5s^2+4)$. Take inverse Laplace transforms to get the answer:

SAGE

```
sage: s,t = var('s t')
sage: inverse_laplace((3*s^3 + 9*s)/(s^4 + 5*s^2 + 4),s,t)
cos(2*t) + 2*cos(t)
sage: inverse_laplace((3*s^3 + 15*s)/(s^4 + 5*s^2 + 4),s,t)
4*cos(t) - cos(2*t)
```

Therefore, $x_1(t) = \cos(2t) + 2\cos(t)$, $x_2(t) = 4\cos(t) - \cos(2t)$. Using SAGE, this can be plotted parametrically using

SAGE

```
sage: P = parametric_plot([cos(2*t) + 2*cos(t),4*cos(t) - cos(2*t)],0,3)
sage: show(P)
```

You can also try

SAGE+Maxima

```
sage.: maxima.plot2d('cos(2*x) + 2*cos(x)','[x,0,1]','[plot_format, openmath]')
```

for the output of a slightly different looking plotting program.



Figure 3: curves $x(t) = cos(2 * t) + 2 * cos(t)$, $y(t) = 4 * cos(t) - cos(2 * t)$ along the t-axis.

Exercise: Solve

$$\begin{cases} x + 2y + z &= 1 \\ -x + 2y - z &= 2 \\ y + 2z &= 3 \end{cases}$$

using (a) row reduction and SAGE, (b) matrix inverses and SAGE.

References

[B] Robert A. Beezer, **A First Course in Linear Algebra**, released under the GNU Free Documentation License, available at `http://linear.ups.edu/`

[CS] Wikipedia article on normal modes of coupled springs: http://en.wikipedia.org/wiki/Normal_mode

[H] Jim Hefferon, **Linear Algebra**, released under the GNU Free Documentation License, available at http://joshua.smcvt.edu/linearalgebra/

[S] W. Stein, SAGE, http://sage.scipy.org/, http://www.sagemath.org/